

Astrowerkstatt : die Programme

Autor(en): **Klaus, G. / Jost-Hediger, H.**

Objektyp: **Article**

Zeitschrift: **Orion : Zeitschrift der Schweizerischen Astronomischen Gesellschaft**

Band (Jahr): **53 (1995)**

Heft 268

PDF erstellt am: **16.04.2024**

Persistenter Link: <https://doi.org/10.5169/seals-898725>

Nutzungsbedingungen

Die ETH-Bibliothek ist Anbieterin der digitalisierten Zeitschriften. Sie besitzt keine Urheberrechte an den Inhalten der Zeitschriften. Die Rechte liegen in der Regel bei den Herausgebern.

Die auf der Plattform e-periodica veröffentlichten Dokumente stehen für nicht-kommerzielle Zwecke in Lehre und Forschung sowie für die private Nutzung frei zur Verfügung. Einzelne Dateien oder Ausdrucke aus diesem Angebot können zusammen mit diesen Nutzungsbedingungen und den korrekten Herkunftsbezeichnungen weitergegeben werden.

Das Veröffentlichen von Bildern in Print- und Online-Publikationen ist nur mit vorheriger Genehmigung der Rechteinhaber erlaubt. Die systematische Speicherung von Teilen des elektronischen Angebots auf anderen Servern bedarf ebenfalls des schriftlichen Einverständnisses der Rechteinhaber.

Haftungsausschluss

Alle Angaben erfolgen ohne Gewähr für Vollständigkeit oder Richtigkeit. Es wird keine Haftung übernommen für Schäden durch die Verwendung von Informationen aus diesem Online-Angebot oder durch das Fehlen von Informationen. Dies gilt auch für Inhalte Dritter, die über dieses Angebot zugänglich sind.



Daraus lässt sich ableiten, dass der Messfehler (ΔV_0) des Veränderlichen (V_0), da nur eine Messung vorliegt, nicht beseitigen lässt. Er muss auf andere Art und Weise abgeschätzt werden. Die Messfehler der Vergleichssterne ($\Delta V_0 - \Delta V_n$) sollten, beim Vorliegen von genügend Messresultaten und bei rein zufälligen Fehlern, mit der Wurzel der Anzahl Messungen abnehmen. Da ich aber nur 5 Vergleichssterne verwendete, entschloss ich mich zur Anwendung einer anderen Methode, um den Fehler zu verkleinern und abzuschätzen.

Ich subtrahiere jeden Vergleichssterne von jedem anderen Vergleichssterne und ermittle dann die Varianz. Es ergibt sich dabei folgendes Bild:

Vergleichssterne	- V2	- V3	- V4	- V5
V1	0.02492	0.0222	0.01557	0.02603
V2		0.00161	0.00248	0.00108
V3			0.02603	0.00183
V4				0.00383

Es zeigt sich, dass die Vergleichssterne V2 und V5 mit einer Varianz von 0,00108 am besten als Messbasis geeignet sind.

Als zweite Methode zur Ermittlung, welche der in Bild 3 ermittelten Lichtkurven am besten geeignet sind, berechne ich die Korrelation zwischen allen Lichtkurven. Es ergibt sich folgendes Bild:

Vergleichssterne	V0 - V2	V0 - V3	V0 - V4	V0 - V5
V0-V1	0.9523	0.9278	0.9579	0.9646
V0-V2		0.9766	0.9818	0.9819
V0-V3			0.9516	0.9789
V0-V4				0.9632

Also auch mit dieser Methode heisst der Sieger Vergleichssterne V2 und V5. Die mit diesen Vergleichssterne ermittelten Kurven nehme ich als gültig und bilde mit dem Mittelwert die endgültige Kurve.

Wie gross ist nun die erzielte Genauigkeit? Sie setzt sich aus dem Fehler der Vergleichssterne und dem Fehler beim Veränderlichen zusammen. Der Mittlere Fehler dürfte nach Auswertung aller Daten ca. 0,006 Magnituden, der maximale Fehler ca. 0,025 Magnituden betragen.

Quellenangaben

- [1] *Sterne und Weltraum*, Ausgabe Oktober 1994
- [2] *Astronomical Photometry*, Henden und Kaitchuck, Verlag Willmann - Bell
- [3] *Bildbearbeitungssoftware*: AIP und Batch-Pix von Richard Berry
- [4] *Photometriesoftware*: CCD-RED und CCD-PHOT von Henden und Kaitchuck

HUGO JOST-HEDIGER
Lingeriz 89, 2540 Grenchen

Astrowerkstatt

Die Programme

G. KLAUS – H. JOST-HEDIGER

Zu den zwei in den Artikel «Jupitercrash» und «DY-PE» erwähnten Berechnungen des «Jupiterzentralmeridians» und des «Heliozentrischen Julianischen Datums» hat Gerhart Klaus die zwei nachfolgenden Programme in Q_BASIC geschrieben. Sie können bei H. Jost, Lingeriz 89, 2540 Grenchen gegen Voreinsendung einer Diskette samt frankiertem Antwortcuvert bezogen werden.

Berechnung des Zentralmeridians von Jupiter

Sprache: Q-BASIC

```

10 CLS: WIDTH 40, 25
20 PRINT " JUPITER ZENTRALMERIDIAN"
30 PRINT "-----"
40 PRINT
50 DEFDBL A-Z
60 RR = 3.141592653589# / 180
200 REM: JULDATE (SKY AND TEL 5/84/454)
210 PRINT " BEOBACHTUNGSZEIT (UT)"
220 INPUT " Jahr,Monat,Tag : ", Y, M, T
230 INPUT " Std,Min : ", U, V
240 D = T + U / 24 + V / 1440

```

```

250 DI = INT(D): F = D - DI - .5
260 J = -INT(7 * (INT((M + 9) / 12) + Y) / 4)
270 S = SGN(M - 9): A = ABS(M - 9)
280 JI = INT(Y + S * INT(A / 7))
290 JI = -INT((INT(JI / 100) + 1) * 3 / 4)
300 J = J + INT(275 * M / 9) + DI + JI
310 J = J + 1721029! + 367 * Y
320 IF F >= 0 THEN 340
330 F = F + 1: J = J - 1
340 JD = F + J
350 DD = JD - 2415020
360 PRINT
400 REM: ZENTRALMERIDIAN (ORION 267 S.89)
410 V = (134.63 + .00111587# * DD) * RR
420 MM = (358.476 + .9856003 * DD) * RR
430 N = (225.328 + .0830853 * DD + .33 * SIN(V)) * RR
440 J = 221.647 + .9025179 * DD - .33 * SIN(V)
450 A = 1.916 * SIN(MM) + .02 * SIN(2 * MM)
460 B = 5.552 * SIN(N) + .167 * SIN(2 * N)
470 K = (J + A - B) * RR
480 R = 1.00014 - .01672 * COS(MM) - .00014 * COS(2 * MM)
490 RI = 5.20867 - .25192 * COS(N) - .0061 * COS(2 * N)
500 DI = SQR(RI * RI + R * R - 2 * RI * R * COS(K))
510 X = R * SIN(K) / DI
520 P = ATN(X / SQR(1 - X * X)) / RR

```



```

530 L1 = 268.28 + 877.8169088# * (DD - DI / 173) + P - B
540 L2 = 290.28 + 870.1869088# * (DD - DI / 173) + P - B
550 C1 = .01016 * (61 + Y - 1995)
560 C2 = .01007 * (61 + Y - 1995)
570 S=L1+C1
580 GOSUB 1000
590 Z1 = S4
600 S = L2 + C2
610 GOSUB 1000
620 Z2 = S4
700 PRINT USING " ZENTRALMERIDIAN I : ###.# "; Z1
710 PRINT USING " ZENTRALMERIDIAN II : ###.# "; Z2
720 PRINT
800 PRINT " Andere Zeit (j/n) ?"
810 Z$ = INKEY$: IF Z$ = "" THEN 810
820 IF Z$ = "j" THEN PRINT: GOTO 230
830 PRINT " Anderes Datum (j/n) ?"
840 Z$ = INKEY$: IF Z$ = "" THEN 840
850 IF Z$ = "j" THEN PRINT: GOTO 220
860 WIDTH 80, 25: END
1000 S1 = S / 360
1010 S2 = FIX(S1)
1020 S3 = S1 - S2
1030 S4 = 360 * S3
1040 RETURN

```

Berechnung des «Heliozentrischen Julianischen Datums»

Sprache: Q-BASIC

```

10 CLS: WIDTH 40, 25
20 PRINT " HELIOZENTRISCHES JULIANISCHES DA-
TUM"
30 PRINT " _____"
40 PRINT
50 DEFDBL A-Z
60 RR = 3.141592653589# / 180
70 EE = 23.45 * RR
100 INPUT " REKTASZENSION (Std,Min,Sek): ", A$, A2, A3
110 GOSUB 1000: RE = A * 15
120 INPUT " DEKLINATION (Grad,Min,Sek): ", A$, A2, A3
130 GOSUB 1000: DE = A: PRINT
200 REM: JULDATE (SKY AND TEL 5/84/454)
210 PRINT " BEOBACHTUNGSZEIT (UT)"
220 INPUT " Jahr, Monat, Tag : ", Y, M, T
230 INPUT " Std, Min : ", U, V
240 D=T+U/24+V/1440
250 D1 = INT(D): F = D - D1 - .5
260 J = -INT(7 * (INT((M + 9) / 12) + Y) / 4)
270 S = SGN(M - 9): A = ABS(M - 9)
280 J1 = INT(Y + S * INT(A / 7))
290 J1 = -INT((INT(J1 / 100) + 1) * 3 / 4)
300 J=J+INT(275 * M/9)+D1+J1
310 J=J+ 1721029! +367 * Y
320 IF F >= 0 THEN 340
330 F=F+ 1: J=J- 1
340 JD=F+J
400 REM: HJD
410 TT = (JD - 2415020) / 36525
420 PP = (1.396041 + .000308 * (TT + .5)) * (TT - .499998)
430 GG = RR * (358.475833# + 35999.04975# * TT - .00015

```

```

* TT * TT)
440 LL = RR * (279.696678# + 36000.76892# * TT + .000303
* TT * TT - PP)
450 XX = .99986 * COS(LL) - .025127 * COS(GG - LL)
460 XX = XX + .008374 * COS(GG - LL)
470 XX = XX + .000105 * COS(2 * GG + LL)
480 XX = XX + .000063 * TT * COS(GG - LL)
490 XX = XX + .000035 * COS(2 * GG - LL)
500 YY = .917308 * SIN(LL) + .023053 * SIN(GG - LL)
510 YY = YY + .007683 * SIN(GG + LL)
520 YY = YY + .000097 * SIN(2 * GG + LL)
530 YY = YY - .000057 * TT * SIN(GG - LL)
540 YY = YY - .000032 * SIN(2 * GG - LL)
550 DT = COS(DE) * COS(RE) * XX
560 DT = DT + (TAN(EE) * SIN(DE) + COS(DE) * SIN(RE))
* YY
570 DT = DT * -.0057755
580 HD=JD+DT
590 COLOR 14, 0: PRINT: PRINT USING " HELIO-JULDAT
= #####.#### "; HD
600 COLOR 7: PRINT: PRINT: PRINT " Andere Zeit (j/n) ?"
610 Z$ = INKEY$: IF Z$ = "" THEN 610
620 IF Z$ = "j" THEN PRINT: GOTO 230
630 PRINT " Anderes Datum (j/n) ?"
640 Z$ = INKEY$: IF Z$ = "" THEN 640
650 IF Z$ = "j" THEN PRINT: GOTO 220
660 PRINT " Andere Koordinaten (j/n) ?"
670 Z$ = INKEY$: IF Z$ = "" THEN 670
680 IF Z$ = "j" THEN PRINT: GOTO 100
690 WIDTH 80, 25: END
1000 REM: BOGENMASS
1010 S = 1: A1 = ABS(VAL(A$))
1020 IF LEFT$(A$, 1) = "-" THEN S = -1
1030 A=S*(A1+A2/60+A3/3600)*RR
1040 RETURN

```

G. KLAUS

Waldeggsstrasse 10, 2540 Grenchen

H. JOST-HEDIGER

Lingeriz 89, 2540 Grenchen

SAG-Rabatt-Katalog «SATURN» für Marken-Teleskope inkl. Selbstbau-Programm «URANUS» gegen Fr. 3.80 in Briefmarken:

*Inbegriffen das gesamte Astro-Programm von
BAADER-PLANETARIUM:*

Refraktoren von Astro-Physics, CCD-Kameras ST4X,
ST5, ST6, ST7, ST8 (in unseren Preisen sind MWST und
Zoll inbegriffen!)

Profitieren Sie vom kostengünstigen 7%-SAG-Rabatt.

Nur Selbstbau-Programm «URANUS» gegen Fr. 1.80 in
Briefmarken. Neu mit Parabolspiegel (ø 6" bis 14"),
Helioskop, Fangspiegel u. -zellen, Hauptspiegelzellen,
SPECTROS-Okulare, Prismen, Okularschlitten, Deklina-
tions- u. Stundenkreise, etc.

Unsere Renner:

Selbstbau-Fernrohr «Saturn» netto Fr. 238.-
Spiegelschleifgarnituren

Schweizerische Astronomische Materialzentrale SAM

Postfach 715 CH 8212 Neuhausen am Rheinfall
Schweiz, Tel 053/22 38 69